
Chimera, Chimera, Unicorn!
In this module, students will be introduced to classifier functions.  In particular, students will 
learn how to teach computers to distinguish between different classes of things without 
telling the computers how to do it, and then test the computers to verify that they learned 
what they could have been expected to learn.  
 

Appropriate for ages 15+.
 

Allow 90+ minutes to complete module

Important note: This module should be led by an instructor with basic Wolfram Language 
knowledge. If you would like to learn the language, please try this free online introduction. If 
you would like a Computational Thinking Initiative ambassador or volunteer to help you run 
an adventure, please contact us.

Learning Objective

◼ Students will learn the basic machine learning principles of training and testing classifiers.

Computational Thinking Principles and Practices

◼ Interpreting a problem or idea in a way that a computer can assist with it.

◼ Simulating things that are hard or impossible to do by real-world experiments.

◼ Treating the computer’s “misunderstandings” as proxies for one’s own.

Standards Alignment

◼ AP Computer Science Principles

◼ LO 1.2.5: Analyze the correctness, usability, functionality and suitability of computational artifacts

◼ LO 3.1.3: Explain the insight and knowledge gained from digitally processed data by using 
appropriate visualizations, notations, and precise language.

◼ LO 3.2.2: Determine how large data sets impact the use of computational processes to discover 
information and knowledge.

◼ LO 5.3.1O: APIs connect software components, allowing them to communicate.

◼ CSTA K–12 Computer Science Standards:

https://www.wolfram.com/language/elementary-introduction/
https://www.wolframcloud.com/objects/computationinitiative/GetInvolved


◼ Level 3A CT 1: Use predefined functions and parameters , classes and methods to divide a complex 
problem into simpler parts.

◼ Level 3A CPP 5: Use Application Programming Interfaces (APIs) and libraries to facilitate 
programming solutions.

◼ Level 3B CT 1: Classify problems as tractable, intractable, or computationally unsolvable.

◼ Level 3B CT 5: Use data analysis to enhance understanding of complex natural and human systems.

Helpful Background

◼ Classifiers are machine tools that classify things: 
https://en.wikipedia.org/wiki/Statistical_classification

◼ Learn more from this video about supervised learning: https://www.wolfram.com/wolfram-
u/catalog/wl031/

◼ Review this guide for more information about creating web apps: 
paclet:tutorial/AdvancedWebFormCreation 

STARTING POINT
“Let’s teach the Wolfram Language to identify small and large numbers! Everyone 
think of some small and some large numbers.”

Collect the suggested numbers into an Association or a List.

In[5]:= bigness = <|"small" → {1, 2, 5}, "large" → {60, 100, 400}|>

Out[5]= small → {1, 2, 5}, large → {60, 100, 400}

“Use the Classify function!”

In[6]:= bignessClassifier = Classify[bigness]

Out[6]= ClassifierFunction
Input type: Numerical
Classes: large, small 

In[10]:= bignessClassifier[3]

Out[10]= small

In[11]:= bignessClassifier[85]

Out[11]= large

◼  What does a ClassifierFunction do?

◼  How about testing some other numbers? 

◼  Is it meaningful to use the classifier to test the values in the training set? Why or why not?

◼  At what point does a small number become a large number, according to the ClassifierFunction?  Is this 
the expected behavior?  How might we tell it to behave that way?  Make sure to talk about how we cannot 
expect classifiers to extrapolate into areas without data in the expected way.

◼  What about adding a “medium” category?

◼  What about negative numbers?  Should numbers like -10^10 be small or large?

◼  Note to students that this could have been done differently; e.g. the students could have manually specified 
that small numbers are numbers less than 20, etc.  This is a simple example, and not everything can be so 
easily specified.

CHECKPOINT
Check to see that students can use a classifier.  

“Let’s teach the Wolfram Language how to classify some mythical creatures!  Why 
don’t we start with getting some images of these creatures?”

Students can choose their own favorite fantasy creatures if they would like.

In[19]:= griffins = WebImageSearch["griffin", "Images", MaxItems → 8]

Out[19]=

In[28]:= {yetis, werewolves} = {WebImageSearch["yeti", "Images", MaxItems → 8],

WebImageSearch["werewolf", "Images", MaxItems → 8]}

Out[28]=

In[32]:= mythicalImages = <|"yeti" → yetis, "werewolf" → werewolves, "griffin" → griffins|>

Out[32]=

If students don’t know about Association objects, collect data in some other format 
that Classify can take.  For example:

In[35]:= mythicalImages =

Thread /@ {griffins → "griffin", yetis → "yeti", werewolves → "werewolf"} // Flatten

Out[35]=

WebImageSearch currently returns 10 results by default (this may be subject to change).  
Students can play around with the MaxItems option, but make sure they don’t end up 
grabbing too few images.

◼  What kinds of data do you think ClassifierFunction would be good at classifying?  Text?  Images?  
Sound?

◼  How much time and effort did using WebImageSearch save over doing all of that manually?

Possible Pitfalls

◼ Sometimes WebImageSearch may miss a result due to weak internet connection and 
return $Failure, or return something unexpected such as a list of images for an animated 
GIF.  To automatically check for that, check that the Head of each search result is in fact 
Image (the Cases function may be helpful).  Students can also remove offending results 
manually (say, with Delete).

◼ WebImageSearch searches the web, so it may occasionally return irrelevant results.  
Students can remove these manually if they would like, but it may not be necessary.

◼ WebImageSearch uses Service Credits, which are provided to users in a limited one-time 
quantity (check yours with $ServiceCreditsAvailable).  It will take quite a few 

searches to burn through them all, but they are not replenished monthly like other credits.  
For more information, check out the Service Credit information page.  If students spend all 
of their Service Credits, they will have to populate their data sets by some other method.  
For example, they can use one of the many supported external services in the Wolfram 
Language such as "Instagram" or "Flickr" (relevant external accounts will be 
required), or they can use a browser to find a webpage containing many relevant images, 
such as a search results page, and use Import with the "Images"  element to get a list of 
images.  If you are planning on introducing external services, make sure to test them out 
beforehand, as external services frequently update their APIs, which may cause 
ServiceConnect to be temporarily unavailable on certain services.  Students can also 
import images one by one manually (by directly pasting or with Import), but that may 
become tedious.

CHECK POINT
Check that the students are all able to obtain images.  

“Now, let’s build a classifier based on our images!”

In[122]:= mythicalClassifier = Classify[mythicalImages]

Out[122]= ClassifierFunction
Input type: Image
Classes: griffin, werewolf, yeti 

“See if the classifier works on different images you might find online. Find some 
images using your browser and use Import to bring them into your notebook”

In[123]:= Import["https://upload.wikimedia.org/wikipedia/en/2/20/Bevan_Crest.jpg"]

Out[123]=

In[124]:= mythicalClassifier[%]

Out[124]= griffin

In[125]:= Import["https://upload.wikimedia.org/wikipedia/commons/2/27/GermanWoodcut1722.jpg"]

Out[125]=

In[46]:= mythicalClassifier[%]

Out[46]= griffin

“How good is our classifier?”

◼  Ask what it means for a ClassifierFunction to be good.  What constitutes a good classifier?  

◼  How might we measure the “goodness” of a ClassifierFunction?

◼  What kinds of mistakes did the classifier make?

◼  What might make a ClassifierFunction better?

Possible Pitfalls

◼ If the students are using the cloud platform with a weak internet connection or a desktop 
platform on a slow computer, pasting images directly into the notebooks may cause them 
to freeze.  Using Import is better than pasting images in these cases.  If notebooks freeze 
for students using the cloud, explain to them that their notebooks are always autosaved (if 
they have not named their notebooks, their work will be in the most recent unnamed 
cloud object). 

CHECK POINT
Check that students are able to create their own ClassifierFunction.  Make sure students remember 
the previous discussion about checking values in the training set.

“To build real-world classifiers, we need to split our ‘known’ data into training data 
and test data.  We use training data to ‘teach’ the classifier, and then we use test data 
to ‘quiz’ the classifier to see how well it learned.”

Students may choose to continue using the mythical creatures they collected in the 
previous section, or they may want to try new data. 

“Let’s start by listing the creatures we want to find. You can use your own list of 
creatures you’d like to classify.”

In[133]:= legendaryCreatures = {"dragon", "unicorn", "centaur", "chimera"}

Out[133]= {dragon, unicorn, centaur, chimera}

“Here we will get a new set of images by making our own function to give us images 
of the creatures in the structure that we need. We can then use that function with the 
list of the types of creatures we just made.”

In[38]:= getImages = Cases[_Image]@WebImageSearch[#, "Images", MaxItems → 16] & // AssociationMap;

In[39]:= legendaryImages = getImages[legendaryCreatures]

Out[39]=

“Now it is time to divide the data into a group of images for training the classifier and 
another group to test whether the classifier is working well. We found 16 images of 
each type of creature, so let’s make the training set the first 8 images of each 
creature. We’ll use the last 8 images of each creature to test the classifier.”

In[50]:= training = #[[ ;; 8]] & /@ legendaryImages;

test = #[[9 ;;]] & /@ legendaryImages;

This code uses Span, Map, and pure anonymous functions, which makes the code look very 
complicated. Check the documentation and chapter 26 of An Elementary Introduction to the 

Wolfram Language if you or the students are unfamiliar with this notation.

◼  Why do we need to separate the training and test images?

◼  Could we use this same approach to classify images of other kinds of things?

◼  What could you do if you had an automatic classifier for things you might see in the real world (instead of 
imaginary creatures)?

◼  Can you name one practical use for an image classifier? Can you name one silly use for an image classifier?

CHECK POINT
Check that students can all split the data into training and test data.

“Classify the data.”

In[153]:= legendaryClassifier = Classify[training]

Out[153]= ClassifierFunction
Input type: Image
Classes: centaur, chimera, dragon, unicorn 

“Instead of just testing individual images like we did before, we can test the 
performance of our classifier across our whole test set of images with the 
ClassifierMeasurements function.”

In[155]:= legendaryTest = ClassifierMeasurements[legendaryClassifier, test]

Out[155]= ClassifierMeasurementsObject
Classifier: LogisticRegression
Number of test examples: 31 

In[156]:= legendaryTest["Accuracy"]

Out[156]= 0.709677

“Accuracy is on a scale of 0 to 1. An accuracy of 0.7 means that the classifier was right 
on 7 examples out of 10 test examples.”

“Like humans, computers may get some things confused more easily than others. For 
instance, it might be harder to tell the difference between a lion and a tiger than 
between a lion and a space shuttle. We can see what our classifier confused by 
making a matrix.”

In[158]:= legendaryTest["ConfusionMatrixPlot"]

Out[158]=

“Where was our classifier most confused? Why might those creatures be confusing?”

“Maybe it would help to look at the exact images that were put into the wrong 
categories. Let’s see the chimeras that our classifier labeled as dragons. Can you see 
why it might have been confused?”

In[199]:= legendaryTest["Examples" → {"chimera", "dragon"}]

Out[199]=  → chimera, → chimera

◼  What does this say about the accuracy of the ClassifierFunction? Does it do better than random 
guessing?  How do we know?

◼  What sorts of images were most likely to fool the ClassifierFunction?  What sorts of images was the 
ClassifierFunction most likely to classify correctly?

◼  How does using more or less training data change how good our ClassifierFunction is?  At what point 
does it become no longer useful to add more data?

◼  How does using more or less testing data change how much we know about how good our ClassifierFunc
tion is?  At what point does it become no longer useful to test on more data?

◼  How do you think the results would change if we had more than five classes in the classifier?

CHECK POINT
Check that students are capable of properly manipulating a ClassifierMeasurementsObject.

“Now that we know that our classifier works reasonably well, let’s make it into a web 
app! This code will deploy a form that you can share with your friends so they can try 
your classifier on their own pictures”

In[74]:= CloudDeploy@FormPage[{{"x", "Creature Image"} → "Image"}, legendaryClassifier[#x] &,

AppearanceRules → <|"Title" → "Identify a Fantasy Creature!", "Description" →

Column@{"Decide whether your creature is one of these:", StringTemplate["• ``"] /@

ClassifierInformation[legendaryClassifier, "Classes"] // Column}|>]

Out[74]= CloudObjecthttps://www.wolframcloud.com/objects/9eeef493-9a0c-4e19-bc5d-1eda54da3d43

“Share your web app with a neighbor!”

◼  What other apps could you make using a classifier? How could you incorporate a classifier as part of a larger 
project?

◼  Think of a friend or family member. What automatic classifier app could you make that would be fun or 
helpful for that person?

◼  Students may be interested in imageidentify.com, which is a whole website built on classifying different 
images.  Very advanced students may also be interested in NetModel["Wolfram ImageIdentify Net 
for WL 11.1"].

CHECK POINT
Check that students can successfully deploy their forms.

FINAL POINT
Ten minutes before end of module time

Summarize

Summarize what was done in the module and talk about findings.

Refer

Refer back to the learning objective and summarize how you have reached it.

Extend

Extend the module to the future. For example, "If you have time at home, try classifying other things.  
Also take a look at ImageIdentify."

Possible Additional Relevant Functions

ClassifierInformation • ClusterClassify • Nearest • FeatureExtraction • 
ImageIdentify

Possible Pitfalls

◼ Students may need a refresher of functions such as Map (/@), Cases, Thread, and 
Function (&).

◼ Searching the web in a classroom setting will always have its challenges.  
WebImageSearch defaults to the option AllowAdultContent→False, but online 
content censoring is never perfect.  While we do not expect students to encounter 
inappropriate material, be cognizant of this fact, and try not to let distracting content get 
in the way of the lesson should it pop up.

